```
%-------------------------------------------------------------------------------
% MEMORY_TEST-1
% VER 1.0
%
% MARCHING MEMORY TEST PROGRAM FOR PERSEUS-8 (CPU:R6502)
%
% MITSURU YAMADA 02/MAY/2021
%
% EXECUTION PROCEDURE
%      (1) TEST FOR DATA BUS
%      (2) MARCHING TEST FOR ADDRESS BUS
%
% RESULT MESSAGE (ADDRESS $0003)
%      RESULT =$00; LAUNCH THIS PROGRAM FAILURE
%             =$01; DATA BUS TEST FAILURE
%             =$02; MARCHING TEST '$00' PATTERN FILL FAILURE
%             =$03; MARCHING TEST '$00' PATTERN READING FAILURE
%             =$04; MARCHING TEST '$FF' PATTERN READING FAILURE
%             =$05; TEST COMPLETE
%
%-------------------------------------------------------------------------------
% COPYRIGHT (C) 2021 MITSURU YAMADA. ALL RIGHTS RESERVED.
%
% ------------------------------------------------------------------------------
% VARIABLES
             SYMBOL                    DATA
             RESULT                    $0003        TEST RESULT
             P0                        $0004        ADDRESS POINTER (16BIT)
             P2                        $0008        END ADDRESS POINTER(16BIT)
%
% ------------------------------------------------------------------------------
%
%                          ADDRESS(HEX) DATA(HEX)
%-------------------------------------------------------------------------------
% PROGRAM START ADDRESS    MARCHING_1   $F820
%
% RESET VECTOR
%             .ORIGIN  $FFFC
                                 FFFC   20 F8
%-------------------------------------------------------------------------------
%
%-------------------------------------------------------------------------------
% MEMORY TEST MAIN PROGRAM
             .ORIGIN  $F820
%
MARCHING_1   CLC                  F820   18           CLEAR CARRY FLAG
             LDX  #$FF            F821   A2 FF        INITIALIZE STACK POINTER
             TXS                  F823   9A
%
             LDX  #$04            F824   A2 04        IX POINT ADDRESS POINTER
             LDY  #$08            F826   A0 08        IY POINT START ADDRESS
%
             LDA  #$01            F828   A9 01        INITIALIZE RESULT
             STA  RESULT          F82A   85 03
%-------------------------------------------------------------------------------
% DATA BUS TEST
%
DATA_TEST    JSR SET_ADDRESS      F82C   20 7B F8
             LDA  #$01            F82F   A9 01        BIT PATTERN FOR DATA BUS TEST
L10          STA  (P0,X)          F831   81 00        WRITE TEST PATTERN
             CMP  (P0,X)          F833   C1 00        READ TEST PATTERN
             BNE  L02             F835   D0 42
             ASL                  F837   0A           SHIFT LEFT TEST PATTERN
             BCC  L10             F838   90 F7
             INC  RESULT          F83A   E6 03        RESULT=$02
             NOP                  F83C   EA
%-------------------------------------------------------------------------------
% MARCHING TEST
%      FILL ZERO PATTERN FROM START ADDRESS TO END ADDRESS
%
FILL_ZERO    LDA  #$00            F83D   A9 00        ZERO PATTERN
             STA  (P0,X)          F83F   81 00
```

```
             JSR  INC_PTX_1      F841   20 8C F8      INCREMENT ADDRESS POINTER
             JSR  CMP_PTX_1      F844   20 93 F8      REACHED END ADDRESS?
             BNE  FILL_ZERO      F847   D0 F4
             INC  RESULT         F849   E6 03         RESULT=$03
%
VERIFY_ZERO  JSR  SET_ADDRESS    F84B   20 7B F8
L03          LDA  (P0,X)         F84E   A1 00         READ BIT ZERO CORRECTLY?
             BNE  L02            F850   D0 27
             LDA  #$FF           F852   A9 FF
             STA  (P0,X)         F854   81 00         WRITE #$FF
             JSR  INC_PTX_1      F856   20 8C F8      INCREMENT ADDRESS POINTER
             JSR  CMP_PTX_1      F859   20 93 F8      REACHED END ADDRESS?
             BNE  L03            F85C   D0 F0
             INC  RESULT         F85E   E6 03         RESULT=$04
%
VERIFY_ONE   JSR  SET_ADDRESS    F860   20 7B F8
L04          LDA  (P0,X)         F863   A1 00         READ BIT ONE CORRECTLY?
             CMP  #$FF           F865   C9 FF
             BNE  L02            F867   D0 10
             LDA  #$00           F869   A9 00
             STA  (P0,X)         F86B   81 00         WRITE #$00
             JSR  INC_PTX_1      F86D   20 8C F8      INCREMENT ADDRESS POINTER
             JSR  CMP_PTX_1      F870   20 93 F8      REACHED END ADDRESS?
             BNE  L04            F873   D0 EE
             INC  RESULT         F875   E6 03         RESULT=$05
L06          BNE  L06            F877   D0 FE         SUCCESSFUL TERMINATION
L02          BNE  L02            F879   D0 FE         ABNORMAL TERMINATION
%
% ------------------------------------------------------------------------------
%
SET_ADDRESS  LDA  #$00           F87B   A9 00         SET START ADDRESS $0200 TO POINTER P0
             STA  P0             F87D   85 04
             LDA  #$02           F87F   A9 02
             STA  P0+1           F881   85 05
             LDA  #$00           F883   A9 00         SET END ADDRESS $2000 TO POINTER P2
             STA  P2             F885   85 08
             LDA  #$20           F887   A9 20
             STA  P2+1           F889   85 09
             RTS                 F88B   60
%
% ------------------------------------------------------------------------------
% INCREMENT 16BIT POINTER  (PARAMETER; 16BIT MEMORY REGISTER POINTED BY IX)
%
INC_PTX_1    INC  $00,X          F88C   F6 00         INCREMENT LOWER BYTE
             BNE  L07            F88E   D0 02
             INC  $01,X          F890   F6 01         INCREMENT UPPER BYTE
L07          RTS                 F892   60
%
% ------------------------------------------------------------------------------
% COMPARE 16BIT POINTER  (PARAMETER; 16BIT MEMORY REGISTERS POINTED BY IX & IY)
%
CMP_PTX_1    LDA  $00,X          F893   B5 00
             CMP  $0000,Y        F895   D9 00 00      COMPARE LOWER BYTE
             BNE  L08            F898   D0 05
             LDA  $01,X          F89A   B5 01
             CMP  $0001,Y        F89C   D9 01 00      COMPARE UPPER BYTE
L08          RTS                 F89F   60
%
% ------------------------------------------------------------------------------
% END OF PROGRAM
```